

NAG Library Routine Document

C09BAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

C09BAF computes the real, continuous wavelet transform in one dimension.

2 Specification

```
SUBROUTINE C09BAF (WAVNAM, WPARAM, N, X, NSCAL, SCALES, C, IFAIL)
  INTEGER          WPARAM, N, NSCAL, SCALES(NSCAL), IFAIL
  REAL (KIND=nag_wp) X(N), C(NSCAL,N)
  CHARACTER(*)     WAVNAM
```

3 Description

C09BAF computes the real part of the one-dimensional, continuous wavelet transform

$$C_{s,k} = \int_{\mathbb{R}} x(t) \frac{1}{\sqrt{s}} \psi^* \left(\frac{t-k}{s} \right) dt,$$

of a signal $x(t)$ at scale s and position k , where the signal is sampled discretely at n equidistant points x_i , for $i = 1, 2, \dots, n$. ψ is the wavelet function, which can be chosen to be the Morlet wavelet, the derivatives of a Gaussian or the Mexican hat wavelet (* denotes the complex conjugate). The integrals of the scaled, shifted wavelet function are approximated and the convolution is then computed.

The mother wavelets supplied for use with this routine are defined as follows.

1. The Morlet wavelet (real part) with nondimensional wave number κ is

$$\psi(x) = \frac{1}{\pi^{1/4}} \left(\cos(\kappa x) - e^{-\kappa^2/2} \right) e^{-x^2/2},$$

where the correction term, $e^{-\kappa^2/2}$ (required to satisfy the admissibility condition) is included.

2. The derivatives of a Gaussian are obtained from

$$\hat{\psi}^{(m)}(x) = \frac{d^m (e^{-x^2})}{dx^m},$$

taking $m = 1, \dots, 8$. These are the Hermite polynomials multiplied by the Gaussian. The sign is then adjusted to give $\hat{\psi}^{(m)}(0) > 0$ when m is even while the sign of the succeeding odd derivative, $\hat{\psi}^{(m+1)}$, is made consistent with the preceding even numbered derivative. They are normalized by the L^2 -norm,

$$p_m = \left(\int_{-\infty}^{\infty} [\hat{\psi}^{(m)}(x)]^2 dx \right)^{1/2}$$

The resulting normalized derivatives can be written in terms of the Hermite polynomials, $H_m(x)$, as

$$\psi^{(m)}(x) = \frac{\alpha H_m(x) e^{-x^2}}{p_m},$$

where

$$\alpha = \begin{cases} 1, & \text{when } m = 0, 3 \text{ mod } 4; \\ -1, & \text{when } m = 1, 2 \text{ mod } 4. \end{cases}$$

Thus, the derivatives of a Gaussian provided here are,

$$\psi^{(1)}(x) = -\left(\frac{2}{\pi}\right)^{1/4} 2xe^{-x^2},$$

$$\psi^{(2)}(x) = -\left(\frac{2}{\pi}\right)^{1/4} \frac{1}{\sqrt{3}}(4x^2 - 2)e^{-x^2},$$

$$\psi^{(3)}(x) = \left(\frac{2}{\pi}\right)^{1/4} \frac{1}{\sqrt{15}}(8x^3 - 12x)e^{-x^2},$$

$$\psi^{(4)}(x) = \left(\frac{2}{\pi}\right)^{1/4} \frac{1}{\sqrt{105}}(16x^4 - 48x^2 + 12)e^{-x^2},$$

$$\psi^{(5)}(x) = -\left(\frac{2}{\pi}\right)^{1/4} \frac{1}{3\sqrt{105}}(32x^5 - 160x^3 + 120x)e^{-x^2},$$

$$\psi^{(6)}(x) = -\left(\frac{2}{\pi}\right)^{1/4} \frac{1}{3\sqrt{1155}}(64x^6 - 480x^4 + 720x^2 - 120)e^{-x^2},$$

$$\psi^{(7)}(x) = \left(\frac{2}{\pi}\right)^{1/4} \frac{1}{3\sqrt{15015}}(128x^7 - 1344x^5 + 3360x^3 - 1680x)e^{-x^2},$$

$$\psi^{(8)}(x) = \left(\frac{2}{\pi}\right)^{1/4} \frac{1}{45\sqrt{1001}}(256x^8 - 3584x^6 + 13440x^4 - 13440x^2 + 1680)e^{-x^2}.$$

3. The second derivative of a Gaussian is known as the Mexican hat wavelet and is supplied as an additional function in the form

$$\psi(x) = \frac{2}{(\sqrt{3}\pi^{1/4})}(1 - x^2)e^{-x^2/2}.$$

The remaining normalized derivatives of a Gaussian can be expressed as multiples of the exponential $e^{-t^2/2}$ by applying the substitution $x = t/\sqrt{2}$ followed by multiplication with the scaling factor, $1/\sqrt[4]{2}$.

4 References

Daubechies I (1992) *Ten Lectures on Wavelets* SIAM, Philadelphia

5 Arguments

1: WAVNAM – CHARACTER(*) *Input*

On entry: the name of the mother wavelet. See the C09 Chapter Introduction for details.

WAVNAM = 'MORLET'

Morlet wavelet.

WAVNAM = 'DGAUSS'

Derivative of a Gaussian wavelet.

WAVNAM = 'MEXHAT'

Mexican hat wavelet.

Constraint: WAVNAM = 'MORLET', 'DGAUSS' or 'MEXHAT'.

- 2: WPARAM – INTEGER *Input*
On entry: the nondimensional wave number for the Morlet wavelet or the order of the derivative for the Gaussian wavelet. It is not referenced when WAVNAM = 'MEXHAT'.
Constraints:
 if WAVNAM = 'MORLET', $5 \leq \text{WPARAM} \leq 20$;
 if WAVNAM = 'DGAUSS', $1 \leq \text{WPARAM} \leq 8$.
- 3: N – INTEGER *Input*
On entry: the size, n , of the input dataset x .
Constraint: $N \geq 2$.
- 4: X(N) – REAL (KIND=nag_wp) array *Input*
On entry: X contains the input dataset $X(j) = x_j$, for $j = 1, 2, \dots, n$.
- 5: NSCAL – INTEGER *Input*
On entry: the dimension of the array SCALES and the first dimension of the array C as declared in the (sub)program from which C09BAF is called. The number of scales to be computed.
Constraint: $\text{NSCAL} \geq 1$.
- 6: SCALES(NSCAL) – INTEGER array *Input*
On entry: the scales at which the transform is to be computed.
Constraint: $\text{SCALES}(i) \geq 1$, for $i = 1, 2, \dots, \text{NSCAL}$.
- 7: C(NSCAL, N) – REAL (KIND=nag_wp) array *Output*
On exit: the transform coefficients at the requested scales, where $C(i, j)$ is the transform coefficient $C_{i,j}$ at scale i and position j .
- 8: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, WAVNAM = $\langle \text{value} \rangle$ was an illegal value.

IFAIL = 2

On entry, WAVNAM = 'MORLET' and WPARAM = $\langle value \rangle$.
 Constraint: if WAVNAM = 'MORLET', $5 \leq WPARAM \leq 20$.

On entry, WAVNAM = 'DGAUSS' and WPARAM = $\langle value \rangle$.
 Constraint: if WAVNAM = 'DGAUSS', $1 \leq WPARAM \leq 8$.

IFAIL = 3

On entry, N = $\langle value \rangle$.
 Constraint: $N \geq 2$.

IFAIL = 5

On entry, NSCAL = $\langle value \rangle$.
 Constraint: $NSCAL \geq 1$.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The accuracy of C09BAF is determined by the fact that the convolution must be computed as a discrete approximation to the continuous form. The input signal, x , is taken to be piecewise constant using the supplied discrete values.

8 Parallelism and Performance

C09BAF is not threaded in any implementation.

9 Further Comments

Workspace is internally allocated by C09BAF. The total size of these arrays is $2^{13} + (N + n_k - 1)$ real elements and n_k integer elements, where $n_k = k \times \max(\text{SCALES}(i))$ and $k = 17$ when WAVNAM = 'MORLET' or 'DGAUSS' and $k = 11$ when WAVNAM = 'MEXHAT'.

10 Example

This example computes the continuous wavelet transform of a dataset containing a single nonzero value representing an impulse. The Morlet wavelet is used with wave number $\kappa = 5$ and scales 1, 2, 3, 4.

10.1 Program Text

```

Program c09bafe

!      C09BAF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
Use nag_library, Only: c09baf, nag_wp
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Integer                     :: ifail, j, n, nscal, wparam
Character(10)               :: wavnam
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: c(:,,:), x(:)
Integer, Allocatable        :: scales(:)
!      .. Intrinsic Procedures ..
Intrinsic                   :: trim
!      .. Executable Statements ..
Write (nout,*) 'C09BAF Example Program Results'
Write (nout,*)
!      Skip heading in data file
Read (nin,*)
!      Read problem parameters
Read (nin,*) n, nscal
Allocate (c(nscal,n),scales(nscal),x(n))
Read (nin,*) wavnam, wparam
Write (nout,99999) trim(wavnam), wparam, n, nscal

!      Read data array and write it out

Read (nin,*) scales(1:nscal)
Read (nin,*) x(1:n)

Write (nout,99998) scales(1:nscal)
Write (nout,99997) x(1:n)

!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
ifail = 0
Call c09baf(wavnam,wparam,n,x,nscal,scales,c,ifail)

Write (nout,99996) nscal
Write (nout,99995) scales(1:nscal)
Do j = 1, n
  Write (nout,99994) c(1:nscal,j)
End Do

99999 Format (2X,'Parameters read from file ::',/,4X,'Wavelet : ',A,      &
', wparam : ',I6,/,10X,'n : ',I6,', nscal : ',I6)
99998 Format (/,2X,'Input Data ::',/,4X,' Scales : ',5(I8,1X),:)
99997 Format (5X,'      x : ',5(F8.3,1X),(/,13X,5(F8.3,1X)),:)
99996 Format (/,2X,'Number of Scales : ',I10)
99995 Format (2X,'Wavelet coefficients C ::',/,4X,'Scale : ',I7,3I13)
99994 Format (10X,4(1P,E11.4,2X),:)
End Program c09bafe

```

10.2 Program Data

```

C09BAF Example Program Data
10      4      : n, nscal
MORLET 5      : wavnam
1      2      3      4      : scales(1:nscal)
0.0 0.0 0.0 0.0 1.0
0.0 0.0 0.0 0.0 0.0 : x(1:n)

```

10.3 Program Results

C09BAF Example Program Results

Parameters read from file ::

```
Wavelet : MORLET, wparam : 5
n       : 10, nscal : 4
```

Input Data ::

```
Scales :      1      2      3      4
x       : 0.000  0.000  0.000  0.000  1.000
          0.000  0.000  0.000  0.000  0.000
```

Number of Scales : 4

Wavelet coefficients C ::

```
Scale :      1      2      3      4
-1.7651E-05  1.5012E-04  5.2331E-02  1.4454E-01
-1.3643E-03 -5.8141E-02  1.7057E-01 -8.4364E-02
 4.6511E-03  1.8442E-01 -1.4891E-01 -2.8870E-01
 8.9294E-02 -2.6380E-01 -2.6822E-01 -9.4993E-02
-9.2563E-02  1.3289E-01  2.5680E-01  2.8293E-01
-9.2563E-02  1.3289E-01  2.5680E-01  2.8293E-01
 8.9294E-02 -2.6380E-01 -2.6822E-01 -9.4993E-02
 4.6511E-03  1.8442E-01 -1.4891E-01 -2.8870E-01
-1.3643E-03 -5.8141E-02  1.7057E-01 -8.4364E-02
-1.7651E-05  1.5012E-04  5.2331E-02  1.4454E-01
```
